

Ohm's Law the Easy Way

By Michael A. Covington

The 36-line BASIC program shown here relates voltage, current, resistance, and power. Given any two of these, it computes the others automatically. Unlike most such programs, this one doesn't require you to say in advance which calculation you want it to do or what formula to use; it examines the number you give it and chooses the right formula automatically.

The BASIC statements used here will run correctly on almost any computer. The program is small enough to run on pocket computers and low-end micros such as the Timex-Sinclair 1000. In the interest of brevity, it does not test for erroneous input (for example, fewer than two values given).

The unknown values are identified by giving them as zero (they would never be zero in a real circuit). On some computers, including the IBM PC, you can simply hit ENTER in response to the question mark for an unknown value, and the computer will behave as if you had typed 0.

The program's strategy is as follows:

1. If voltage was not given, compute it. (Any of three formulas may be used depending on what information is available.)

2. If current was not given, compute it. (Since voltage is now known, there are only two formulas to choose between.)

```

10 REM Ohm's law program
20 REM Michael A. Covington
30 PRINT "This program relates voltage, "
40 PRINT "resistance, current, and power."
50 PRINT
60 PRINT "Specify any two; type 0 for the"
70 PRINT "ones you want calculated."
80 PRINT
90 PRINT "Voltage (volts):",
100 INPUT E
110 PRINT "Current (amps):",
120 INPUT I
130 PRINT "Resistance (ohms):",
140 INPUT R
150 PRINT "Power (watts):",
160 INPUT P
170 IF E<>0 THEN 210
180 IF I=0 THEN E=SQR(P*R)
190 IF R=0 THEN E=P/I
200 IF P=0 THEN E=I*R
210 IF I<>0 THEN 240
220 IF R=0 THEN I=P/E
230 IF R<>0 THEN I=E/R
240 IF R=0 THEN R=E/I
250 IF P=0 THEN P=E*I
260 PRINT
270 PRINT "Voltage (volts):",E
280 PRINT "Current (amps):",I
290 PRINT "Resistance (ohms):",R
300 PRINT "Power (watts):",P
310 PRINT
320 PRINT "Another? Type 1 for yes, 2 for no..."
330 INPUT R
340 PRINT
350 IF R=1 THEN 90
360 END

```

3. Compute resistance and power (from voltage and current, which are known by now).

You can apply the same strategy to any equation that you frequently need to solve for any of several va-

riables. There is no need to rummage through a list of formulas, or even choose one from a menu on the screen. After all, the computer doesn't mind doing the extra work, so why not let it? **ME**